```
" DYNAMIC FORCE ANALYSIS VIA CONTOUR METHOD "

Apply [Clear, Names["Global`*"] ] ;

Off[General::spell];
Off[General::spell1];

(* Input data *)
n = 50 ; (* rpm *)
omega = n N[Pi]/30 ; (* rad/s *)

rule = {AB->0.14, AC->0.06, AE->0.25, CD->0.15, FD->0.4, EG->0.5, h->0.01, d->0.001,
hSlider->0.02, wSlider->0.05, rho->8000, g->9.807, Me->-100., phi[t]->N[Pi]/6, phi'[t]-
>omega, phi''[t]->0} ;

(* Position analysis *)

(* Position of joint A *)
xA = yA = 0;
rA = { xA, yA, 0};
Print["rA = ", rA, " [m]"] ;

(* Position of joint C *)
xC = 0 ;
yC = AC ;
rC = { xC, yC, 0} ;
Print["rC = AC = ", rC/.rule, " [m]"] ;

(* Position of joint E *)
xE = 0 ;
yE = -AE ;
rE = { xE, yE, 0};
Print["rE = AE = ", rE/.rule, " [m]"] ;


(* Position, velocity and acceleration of joint B *)
xB = AB Cos [ phi[t] ] ;
yB = AB Sin [ phi[t] ] ;
rB = { xB, yB, 0} ;
Print["rB = AB = ", rB/.rule, " [m]"] ;
vB = D[rB,t] ;
aB = D[D[rB,t],t] ;

(* Position, velocity and acceleration of joint D *)

(* Parameters m and n of line BC: y = m x + n *)
mBC = ( yB - yC ) / ( xB - xC ) ;
nBC = yB - mBC xB ;
eqn41 = ( xDsol - xC )^2 + ( yDsol - yC )^2 - CD^2 == 0 ;
eqn42 = yDsol - mBC xDsol - nBC == 0 ;
solutionD = Solve [ { eqn41 , eqn42 } , { xDsol , yDsol } ];
(* Two solutions for D *)
xD1 = xDsol /. solutionD[[1]];
yD1 = yDsol /. solutionD[[1]];
xD2 = xDsol /. solutionD[[2]];
yD2 = yDsol /. solutionD[[2]];
(* Select the correct position for D *)
If[ (xD1/.rule)<=xC, xD=xD1; yD=yD1, xD=xD2; yD=yD2 ] ;
rD = { xD, yD, 0} ;
Print["rD = AD =", rD/.rule, " [m]"] ;
vD = D[rD,t] ;
aD = D[D[rD,t],t] ;
```

```
(* Angular velocity and acceleration of the link 1 *)
alpha1 = {0, 0, phi''[t]} ;

(* Angular velocity and acceleration of the link 2 and link 3 *)
phi2 = ArcTan[ mBC ] ;
alpha2 = {0, 0, D[D[phi2,t],t]} ;
phi3 = phi2 ;
alpha3 = alpha2 ;

(* Angular velocity and acceleration of the link 4 and link 5 *)

phi4 = ArcTan[(yD-yE)/(xD-xE)] + N[Pi] ;
alpha4 = {0, 0, D[D[phi4,t],t]} ;
phi5 = phi4 ;
alpha5 = alpha4 ;

(* ------------------------- *)
(* Inertia forces and moments *)
(* ------------------------- *)

(* Link 1 *)
m1 = rho AB h d /.rule ;
rC1 = rB/2 ;
Print["rC1 = AC1 =", rC1/.rule, " [m]"] ;
vC1 = vB/2 ;
aC1 = aB/2 ;
Fin1 = - m1 aC1 /.rule ;
G1 = {0, -m1*g, 0} /.rule ;
F1 = ( Fin1 + G1 ) /.rule ;
IC1 = m1 (AB^2+h^2)/12 /.rule ;
M1 = Min1 = - IC1 alpha1 /.rule ;
Print["F1 = ", F1, " [N]"] ;
Print["M1 = ", M1, " [Nm]"] ;

(* Link 2 *)
m2 = rho hSlider wSlider d /.rule ;
rC2 = rB ;
Print["rC2 = AC2 = AB = ", rC2/.rule, " [m]"] ;
vC2 = vB ;
aC2 = aB ;
Fin2 = - m2 aC2 /.rule ;
G2 = {0, -m2*g, 0} /.rule ;
F2 = ( Fin2 + G2 ) /.rule ;
IC2 = m2 (hSlider^2+wSlider^2)/12 /.rule ;
M2 = Min2 = - IC2 alpha2 /.rule ;
Print["F2 = ", F2, " [N]"] ;
Print["M2 = ", M2, " [Nm]"] ;

(* Link 3 *)
m3 = rho FD h d /.rule ;
xC3 = xC + (FD/2-CD) Cos [ phi3 ] ;
yC3 = yC + (FD/2-CD) Sin [ phi3 ] ;
rC3 = { xC3, yC3, 0 } ;
Print["rC3 = AC3 = ", rC3/.rule, " [m]"] ;
vC3 = D[rC3,t] ;
aC3 = D[D[rC3,t],t] ;
Fin3 = - m3 aC3 /.rule ;
G3 = {0, -m3*g, 0} /.rule ;
F3 = ( Fin3 + G3 ) /.rule ;
IC3 = m3 (FD^2+h^2)/12 /.rule ;
M3 = Min3 = - IC3 alpha3 /.rule ;
Print["F3 = ", F3, " [N]"] ;
Print["M3 = ", M3, " [Nm]"] ;

(* Link 4 *)
```

```
m4 = rho hSlider wSlider d /.rule ;
rC4 = rD ;
Print["rC4 = AC4 = AD =", rC4/.rule, " [m]"] ;
vC4 = vD ;
aC4 = aD ;
Fin4 = - m4 aC4 /.rule ;
G4 = {0, -m4*g, 0} /.rule ;
F4 = ( Fin4 + G4 ) /.rule ;
IC4 = m4 (hSlider^2+wSlider^2)/12 /.rule ;
M4 = Min4 = - IC4 alpha4 /.rule ;
Print["F4 = ", F4, " [N]"] ;
Print["M4 = ", M4, " [Nm]"] ;

(* Link 5 *)
m5 = rho EG h d /.rule ;
xC5 = EG/2 Cos [ phi5 ] ;
yC5 = EG/2 Sin [ phi5 ] ;
rC5 = { xC5, yC5, 0} ;
Print["rC5 = AC5 = ", rC5/.rule, " [m]"] ;
vC5 = D[rC5,t] ;
aC5 = D[D[rC5,t],t] ;
Fin5 = - m5 aC5 /.rule ;
G5 = {0, -m5*g, 0} /.rule ;
F5 = ( Fin5 + G5 ) /.rule ;
IC5 = m5 (EG^2+h^2)/12 /.rule ;
M5 = Min5 = - IC5 alpha5 /.rule ;
M5e = { 0, 0, Me } /.rule ;
Print["F5 = ", F5, " [N]"] ;
Print["M5 = ", M5, " [Nm]"] ;

(* --------------- *)
(* Joint reactions *)
(* --------------- *)

(*** Contour 0-3-4-5-0 ***)

" Joint E_R: F05 "

F05Sol = { F05xSol, F05ySol, 0 } ;

"D_T: ∑F(5).DE=0: F05+F5=0 (1)"
rDE = ( rE - rD ) /.rule ;
eqER1 = (F5+F05Sol).rDE == 0 ;

"D_R: ∑M(4&5)_D=0: DExF05+DC5xF5+M4+M5+M5e=0 (2)"
rDC5 = ( rC5 - rD ) /.rule ;
eqER2 = (Cross[rDE,F05Sol]+Cross[rDC5,F5]+M4+M5+M5e)[[3]] == 0 ;
solF05 = Solve[{eqER1,eqER2},{F05xSol,F05ySol}] ;
F05x = F05xSol /.solF05[[1]] ;
F05y = F05ySol /.solF05[[1]] ;
F05 = { F05x, F05y, 0 } ;
Print["Eqs.(1)(2) =>"] ;
Print["F05 = ", F05, " [N]"] ;

" Joint D_T: F45=-F54 at P{xP,yP,0} "

F45Sol = { F45xSol, F45ySol, 0 } ;
F54Sol = - F45Sol ;
rPSol = { xPSol, yPSol, 0 } ;

"E_R: ∑M(5)_E=0: EPxF45+EC5xF5+M5+M5e=0 (3)"
rEP = ( rPSol - rE ) /.rule ;
rEC5 = ( rC5 - rE ) /.rule ;
eqDT1 = (Cross[rEP,F45Sol]+Cross[rEC5,F5]+M5+M5e)[[3]] == 0 ;
```

```
"D_R: ∑M(4)_D=0: DPxF54+M4=0 (4)"
rDP = ( rPSol - rD ) /.rule ;
eqDT2 = (Cross[rDP,F54Sol]+M4)[[3]] == 0 ;

"F45 perpendicular to DE: F45.DE=0 (5)"
eqDT3 = F45Sol.rDE == 0;

"P{xP,yP,0} on DE: (yD-yE)/(xD-xE)=(yP-yE)/(xP-xE) (6)"
eqDT4 = ( (yD-yE)/(xD-xE)/.rule ) == ( (yPSol-yE)/(xPSol-xE)/.rule ) ;

solF45 = Solve[{eqDT1,eqDT2,eqDT3,eqDT4},{F45xSol,F45ySol,xPSol,yPSol}] ;
F45x = F45xSol/.solF45[[1]] ;
F45y = F45ySol/.solF45[[1]] ;
F45 = { F45x, F45y, 0 } ;
F54 = - F45 ;
xP = xPSol/.solF45[[1]] ;
yP = yPSol/.solF45[[1]] ;
rP = { xP, yP, 0 } ;
Print["Eqs.(3)(4)(5)(6)=>"] ;
Print["F45 = ", F45," [N]"] ;
Print["rP = ", rP, " [m]"] ;

" Joint D_R: F34"

F34Sol = { F34xSol, F34ySol, 0 } ;
F43Sol = - F34Sol ;

"D_T: ∑F(4).DE=0: (F4+F34).DE=0 (7)"
rED = ( rD - rE ) /.rule ;
eqDR1 = (F4+F34Sol).rED == 0 ;

"E_R: ∑M(4&5)_E=0: EC4xF4+EC5xF5+EDxF34+M4+M5+M5e=0 (8)"
rEC5 = ( rC5 - rE ) /.rule ;
rEC4 = ( rC4 - rE ) /.rule ;
eqDR2 = (Cross[rEC4,F4]+Cross[rEC5,F5]+Cross[rED,F34Sol]+M4+M5+M5e)[[3]] == 0 ;

solF34 = Solve[{eqDR1,eqDR2},{F34xSol,F34ySol}] ;
F34x = F34xSol/.solF34[[1]] ;
F34y = F34ySol/.solF34[[1]] ;
F34 = { F34x, F34y, 0 } ;
F43 = - F34 ;
Print["Eqs.(7)(8)=>"] ;
Print["F34 = ", F34, " [N]"] ;

(*** Contour 0-1-2-3-0 ***)

" Joint C_R: F03 "

F03Sol = { F03xSol, F03ySol, 0 } ;

"B_T: ∑F(3).CD=0: (F03+F43+F3).CD=0 (9)"
rCD = ( rD - rC ) /.rule ;
eqCR1 = (F03Sol+F43+F3).rCD == 0 ;
"B_R: ∑M(3&2)_B=0: BC3xF3+BCxF03+BDxF43+M2+M3=0 (10)"
rBC3 = ( rC3 - rB ) /.rule  ;
rBC = ( rC - rB ) /.rule  ;
rBD = ( rD - rB ) /.rule  ;
eqCR2 = (Cross[rBC3,F3]+Cross[rBC,F03Sol]+Cross[rBD,F43]+M2+M3)[[3]] == 0 ;

solF03 = Solve[{eqCR1,eqCR2},{F03xSol,F03ySol}] ;
F03x = F03xSol/.solF03[[1]] ;
F03y = F03ySol/.solF03[[1]] ;
F03 = { F03x, F03y, 0 } ;
Print["Eqs.(9)(10)=>"];
Print["F03 = ", F03, " [N]"] ;
```

```
" Joint B_T: F23=-F32 at Q{xQ,yQ,0}"

F23Sol = { F23xSol, F23ySol, 0 } ;
F32Sol = - F23Sol ;
rQSol = { xQSol, yQSol, 0 } ;

"C_R: ∑M(3)_C=0, CQxF23+CC3xF3+CDxF43+M3=0 (11)"
rCQ = ( rQSol - rC ) /.rule ;
rCC3 = ( rC3 - rC ) /.rule ;
rCD = ( rD - rC ) /.rule ;
eqBT1 = (Cross[rCQ,F23Sol]+Cross[rCC3,F3]+Cross[rCD,F43]+M3)[[3]] == 0 ;

"B_R: ∑ M(2)_B=0, BQxF32+M2=0 (12)"
rBQ = ( rQSol - rB ) /.rule ;
eqBT2 = (Cross[rBQ,F32Sol]+M2)[[3]] == 0 ;

"F23 perpendicular to BC: F23.BC=0 (13)"
eqBT3 = ( (yC-yB)/(xC-xB)/.rule ) == - F23xSol/F23ySol ;
"Q{xQ,yQ,0} on BC: (yC-yB)/(xC-xB)=(yC-yQ)/(xC-xQ) (14)"
eqBT4 = ( (yC-yB)/(xC-xB)/.rule ) == ( (yC-yQSol)/(xC-xQSol)/.rule ) ;

solF23 = Solve[{eqBT1,eqBT2,eqBT3,eqBT4},{F23xSol,F23ySol,xQSol,yQSol}] ;
F23x = F23xSol/.solF23[[1]] ;
F23y = F23ySol/.solF23[[1]] ;
F23 = { F23x, F23y, 0 } ;
F32 = - F23 ;
xQ = xQSol/.solF23[[1]] ;
yQ = yQSol/.solF23[[1]] ;
rQ = { xQ, yQ, 0 } ;
Print["Eqs.(11)(12)(13)(14)=>"] ;
Print["F23 = ", F23," [N]"] ;
Print["rQ = ", rQ, " [m]"] ;

" Joint B_R: F12=-F21 "

F12Sol = { F12xSol, F12ySol, 0 } ;
F21Sol = - F12Sol ;

"B_T: ∑F(2).BC=0: (F12+F2).BC=0 (15)"
rBC = ( rC - rB ) /.rule ;
eqBR1 = (F12Sol+F2).rBC == 0 ;

"C_R: ∑M(2&3)_C=0: CBxF12+CC2xF2+CC3xF3+CDxF43+M2+M3=0 (16)"
rCB = ( rB - rC ) /.rule ;
rCC2 = ( rC2 - rC ) /.rule ;
rCC3 = ( rC3 - rC ) /.rule ;
rCD = ( rD - rC ) /.rule ;
eqBR2 = (Cross[rCB,F12Sol]+Cross[rCC2,F2]+Cross[rCC3,F3]+Cross[rCD,F43]+M2+M3)[[3]] ==
0 ;

solF12 = Solve[{eqBR1,eqBR2},{F12xSol,F12ySol}] ;
F12x = F12xSol/.solF12[[1]] ;
F12y = F12ySol/.solF12[[1]] ;
F12 = { F12x, F12y, 0 } ;
F21 = - F12 ;
Print["Eqs.(15)(16)=>"] ;
Print["F12 = ", F12," [N]"] ;

" Motor torque Mm "
M1mSol = { 0, 0, MmSol };
"A_R: ∑M(1)_A=0: ABxF21+AC1xF1+M1+Mm=0 (17)"
rAB = rB /.rule ;
rAC1 = rC1 /.rule ;
eqMA = (Cross[rAB,F21]+Cross[rAC1,F1]+M1+M1mSol)[[3]] == 0 ;
```

```
solMm = Solve[eqMA,MmSol] ;
Mm = MmSol/.solMm[[1]] ;
M1m = { 0, 0, Mm } ;
Print["Eqs.(17)=>"] ;
Print["Mm = ", Mm," [Nm]"] ;

" Joint A_R: F01"

F01Sol = { F01xSol, F01ySol, 0 } ;

"B_R: ∑M(1)_B=0: BAxF01+BC1xF1+M1+Mm=0 (18)"

rBA = -rB /.rule ;
rBC1 = ( rC1 - rB ) /.rule ;
eqAR1 = (Cross[rBA,F01Sol]+Cross[rBC1,F1]+M1+M1m)[[3]] == 0 ;

"B_T: ∑F(1&2).BC=0: (F01+F1+F2).BC (19)"
eqAR2 = (F01Sol+F1+F2).rBC == 0 ;
solF01 = Solve[{eqAR1,eqAR2},{F01xSol,F01ySol}] ;

F01x = F01xSol/.solF01[[1]] ;
F01y = F01ySol/.solF01[[1]] ;
F01 = { F01x, F01y, 0 } ;
Print["Eqs.(18)(19)=>"] ;
Print["F01 = ", F01," [N]"] ;
```

```
 DYNAMIC FORCE ANALYSIS VIA CONTOUR METHOD

rA = {0, 0, 0} [m]

rC = AC = {0, 0.06, 0} [m]

rE = AE = {0, -0.25, 0} [m]

rB = AB = {0.121244, 0.07, 0} [m]

rD = AD ={-0.149492, 0.0476701, 0} [m]

rC1 = AC1 ={0.0606218, 0.035, 0} [m]

F1 = {0.0186142, -0.0990915, 0} [N]

M1 = {0, 0, 0} [Nm]

rC2 = AC2 = AB = {0.121244, 0.07, 0} [m]

F2 = {0.0265917, -0.0631033, 0} [N]

M2 = {0, 0, -0.000028165} [Nm]

rC3 = AC3 = {0.0498308, 0.06411, 0} [m]

F3 = {0.0492489, -0.33315, 0} [N]

M3 = {0, 0, -0.00621962} [Nm]

rC4 = AC4 = AD ={-0.149492, 0.0476701, 0} [m]

F4 = {-0.0369367, -0.0639614, 0} [N]

M4 = {0, 0, 0.0000111583} [Nm]

rC5 = AC5 = {-0.112198, 0.223409, 0} [m]
```

F5 = {-0.0553516, -0.410666, 0} [N]

M5 = {0, 0, 0.00481155} [Nm]

Joint E_R: F05

D_T: ∑F(5).DE=0: F05+F5=0 (1)

D_R: ∑M(4&5)_D=0: DExF05+DC5xF5+M4+M5+M5e=0 (2)

Eqs.(1)(2) =>

F05 = {268.127, 135.039, 0} [N]

Joint D_T: F45=-F54 at P{xP,yP,0}

E_R: ∑M(5)_E=0: EPxF45+EC5xF5+M5+M5e=0 (3)

D_R: ∑M(4)_D=0: DPxF54+M4=0 (4)

F45 perpendicular to DE: F45.DE=0 (5)

P{xP,yP,0} on DE: (yD-yE)/(xD-xE)=(yP-yE)/(xP-xE) (6)

Eqs.(3)(4)(5)(6)=>

F45 = {-268.072, -134.628, 0} [N]

rP = {-0.149492, 0.0476701, 0} [m]

Joint D_R: F34

D_T: ∑F(4).DE=0: (F4+F34).DE=0 (7)

E_R: ∑M(4&5)_E=0: EC4xF4+EC5xF5+EDxF34+M4+M5+M5e=0 (8)

Eqs.(7)(8)=>

F34 = {-268.035, -134.564, 0} [N]

Joint C_R: F03

B_T: ∑F(3).CD=0: (F03+F43+F3).CD=0 (9)

B_R: ∑M(3&2)_B=0: BC3xF3+BCxF03+BDxF43+M2+M3=0 (10)

Eqs.(9)(10)=>

F03 = {-256.71, -272.141, 0} [N]

Joint B_T: F23=-F32 at Q{xQ,yQ,0}

C_R: ∑M(3)_C=0, CQxF23+CC3xF3+CDxF43+M3=0 (11)

B_R: ∑ M(2)_B=0, BQxF32+M2=0 (12)

F23 perpendicular to BC: F23.BC=0 (13)

Q{xQ,yQ,0} on BC: (yC-yB)/(xC-xB)=(yC-yQ)/(xC-xQ) (14)

Eqs.(11)(12)(13)(14)=>

F23 = {-11.3747, 137.91, 0} [N]

rQ = {0.121243, 0.07, 0} [m]

Joint B_R: F12=-F21

B_T: ∑F(2).BC=0: (F12+F2).BC=0 (15)

C_R: ∑M(2&3)_C=0: CBxF12+CC2xF2+CC3xF3+CDxF43+M2+M3=0 (16)

Eqs.(15)(16)=>

F12 = {-11.4013, 137.974, 0} [N]

Motor torque Mm

A_R: ∑M(1)_A=0: ABxF21+AC1xF1+M1+Mm=0 (17)

Eqs.(17)=>

Mm = 17.5332 [Nm]

Joint A_R: F01

B_R: ∑M(1)_B=0: BAxF01+BC1xF1+M1+Mm=0 (18)

B_T: ∑F(1&2).BC=0: (F01+F1+F2).BC (19)

Eqs.(18)(19)=>

F01 = {-11.4199, 138.073, 0} [N]